

# CPE 325: Embedded Systems Laboratory

## Laboratory #4 Tutorial

### Introduction to MSP430 Assembly Language Programming

**Aleksandar Milenković**

Email: [mlenka@uah.edu](mailto:mlenka@uah.edu)

Web: <http://www.ece.uah.edu/~mlenka>

#### Objective:

This tutorial will help you get started with MSP30 assembly program development. In this lab, you will learn the following topics:

*Assembly language programming*

*Creating an application project using assembly language programming*

*Debugging the application using CCS Debug*

#### Notes:

All previous tutorials introducing the TI Experimenter's Board and the Code Composer Studio software development environment are required for successful completion of this lab. Class materials required are: [MSP430 ISA](#) and [MSP430 Assembly Language Programming](#).

#### Contents:

|     |   |   |
|-----|---|---|
| 1   | Problem Statement: Count Number of Characters 'E' in a String ..... | 2 |
| 2   | Counting Characters .....   | 2 |
| 2.1 | Assembly Program.....   | 3 |
| 2.2 | Creating an Application Project .....                               | 5 |
| 2.3 | Program Simulation and Debugging .....                              | 6 |
| 3   | References .....  | 7 |

## 1 Problem Statement: Count Number of Characters 'E' in a String

This section defines the problem that will be solved by the Count Characters program using the MSP430 assembly language. Our task is to develop an assembly program that will scan a given string of characters, for example, "HELLO WORLD, I AM THE MSP430!", and find the number of appearances of the character 'E' in the string. A counter that records the number of characters 'E' is then written to the parallel port P1. The port should be configured as an output port, and the binary value of the port will correspond to the counter value.

Step 1. Analyze the assignment.

To solve this assignment, let us first analyze the problem statement. Your task is to write an assembly program that will count the number of characters 'E' in a string. First, the problem implies that we need to allocate space in memory that will keep the string "HELLO WORLD, I AM THE MSP430!". The string has 29 characters and they are encoded using the ASCII table. Two single quotes at the end allocate a NULL character (0x00). To allocate and initialize a string in memory we can use an assembly language directive `.string` (or `.byte`).

```
myStr    .string "HELLO WORLD, I AM THE MSP430!", ''
```

A label `myStr` marks the beginning of this string in memory, and the value of this symbol corresponds to the starting address of the string. When assembler sees the `.string` directive, it will allocate the space in memory required for the string that follows and initialize the allocated space with corresponding ASCII characters. Including the ASCII NULL character at the end of the string (`ascii(NULL) = 0x00`), the total number of bytes occupied by this directive is 30.

Step 2. Develop a plan.

Our task is now to write a small program that will scan the string, character by character, check whether the current character is equal to the character 'E', and if yes, increment a counter. The string scan is done in a program loop. The program ends when we reach the end of the string, which is detected when the current character is equal to the NULL character (0x00). To scan the string, we will use a register to point to the current character in the string. This pointer register is initialized at the beginning of the program to point to the first character in the string. The pointer will be incremented in each iteration of the program loop. Another register, initialized to zero at the beginning, will serve as the counter, and it is incremented every time the current character is 'E'. After we exit the program loop, the current value of the counter will be written to the port P1, which should be initialized as an output port.

**NOTE:** It is required that you are familiar with the MSP430 instruction set and addressing modes to be able to solve this problem. Also, we will assume that the string is no longer than 255 characters, so the result can be displayed on an 8-bit port.

## 2 Counting Characters

This section will go more in depth into how to write the assembly code to solve this problem in addition to how to build and link the code for the MSP430.

## 2.1 Assembly Program

The assembly code for this program can be seen on the next page. However, here is a brief description of how the code works.

The comments in a single line start with a column character (;). Multi-line comments are not supported. Note the proceeding (-) characters are used to indicate different sections of the assembly file and are good practice for clean readable code. In line 12, you will notice an assembly pre-processor directive that specifies a header file to be included in the source code. The header file includes all macro definitions, for example, special function register addresses (WDTCTL), and control bits (WDTPW+WDTHOLD). Next, in line 18 we allocate the string that is going to be counted. This directive will allocate 30 bytes in memory and initialize it with the string content, placing the ASCII codes for the string characters in the memory as illustrated below.

| Address  | Memory[7:0] |
|----------|-------------|
| myStr    | 'H'         |
| myStr+1  | 'E'         |
| myStr+2  | 'L'         |
| myStr+3  | 'O'         |
| myStr+4  | ' '         |
| myStr+5  | 'W'         |
| myStr+6  | 'O'         |
| myStr+7  | 'R'         |
| ...      |             |
| myStr+28 | '!'         |
| myStr+29 | 0x00 (NULL) |

So how does the program execute on an MSP430? Upon powering-up the MSP430 control logic always generates a reset interrupt request (it is the highest priority interrupt request). The value stored at the address 0xFFFF (the last word in the 64KB address space) is reserved to keep the starting address of the reset handler (interrupt service routine), and the first thing that the microcontroller does is to fetch the content from this address and put it in the program counter (PC, register R0). Thus, the starting address of our program corresponding to the entry point called RESET should be stored at location 0xFFFF. Line 62 declares the reset vector that is then initialized with the value that corresponds to the symbol RESET, which is the starting address of our program.

The first instruction on line 31 initializes the stack pointer register. However, our program does not use the stack, so we can ignore this instruction for now. The instruction on line 32 will set certain control bits of the watchdog timer control register (WDTCTL) to disable it. The watchdog timer by default is active upon reset, generating periodic system resets if left unattended. As this functionality is not needed in our program, we need to disable it.

Parallel ports in the MSP430 microcontroller can be configured as either input or output ports. A control register PxDIR determines whether the port is an input or an output port (we can configure each individual port pin). If you want to enable outputs to be inspected from outside on port pins, you would configure P1DIR register for output direction P1DIR=0xFF. However, in our case we are not interested in driving external port pins as we cannot easily observe their state (without probing outputs). In addition, by doing manual inspection using probes you may cause shorts that may damage your board. Thus, the line of code to set P1DIR direction to output is commented out (the default state of P1DIR is 0x00 which indicates input direction). Still we will write our result to register P1OUT associated to P1, but its content will not be connected to the output port pins. Register R4 is loaded to point to the first character in the string. Register R5, the counter, is cleared before starting the main program loop.

The loop starts at the next label called gnext. We use the autoincrement addressing mode to read a new character (one byte) from the string as seen in line 40. The current character is kept in register R6. We then compare the current character with the NULL character on line 41. If it is the NULL character, the end of the string has been reached and we exit the loop (JEQ lend). Pay attention that we used JEQ instruction. Why? If it is not the end of the string, we compare the current character with 'E'. If there is no match we go back to the first instruction in the loop. Otherwise, we increase the value of the counter (register R5). Finally, once the end of the string has been reached, we move the lower byte from R5 to the P1OUT register on line 48.

```

1  ;-----
2  ; File      : Lab4_D1.asm (CPE 325 Lab4 Demo code)
3  ; Function   : Counts the number of characters E in a given string
4  ; Description: Program traverses an input array of characters
5  ;           : to detect a character 'E'; exits when a NULL is detected
6  ; Input      : The input string is specified in myStr
7  ; Output     : The port P1OUT displays the number of E's in the string
8  ; Author     : A. Milenkovic, milenkovic@computer.org
9  ; Date       : August 14, 2008;
10 ;            : August 5, 2020 (revised)
11 ;-----
12      .cdecls C,LIST,"msp430.h"          ; Include device header file
13
14 ;-----
15      .def      RESET                    ; Export program entry-point to
16                                         ; make it known to linker.
17
18 myStr: .string "HELLO WORLD, I AM THE MSP430!", ''
19      ; .string does add NULL at the end of the string;
20      ; '' ensures that a NULL follows the string.
21      ; You can alternatively use .cstring "HELLO WORLD, I AM THE MSP430!"
22      ; that adds a NULL character at the end of the string automatically.
23 ;-----
24      .text                               ; Assemble into program memory.
25      .retain                             ; Override ELF conditional linking
26                                         ; and retain current section.
27      .retainrefs                         ; And retain any sections that have
28                                         ; references to current section.
29

```

```

30 ;-----
31 RESET:  mov.w  #__STACK_END,SP      ; Initialize stack pointer
32         mov.w  #WDTPW|WDTHOLD,&WDTCTL ; Stop watchdog timer
33
34 ;-----
35 ; Main loop here
36 ;-----
37 main:   ; bis.b  #0FFh,&P1DIR      ; do not output the result on port pins
38         mov.w  #myStr, R4         ; load the starting address of the string
39 into R4
40
41 gnext:  clr.b   R5                 ; register R5 will serve as a counter
42         mov.b  @R4+, R6           ; get a new character
43         cmp    #0,R6              ; is it a null character
44         jeq    lend              ; if yes, go to the end
45         cmp.b  #'E',R6           ; is it an 'E' character
46         jne    gnext             ; if not, go to the next
47         inc.w  R5                 ; if yes, increment counter
48         jmp    gnext             ; go to the next character
49
50 lend:   mov.b  R5,&P1OUT           ; write result in P1OUT (not visible on port
51 pins)
52         bis.w  #LPM4,SR           ; LPM4
53         nop                    ; required only for Debugger
54
55 ;-----
56 ; Stack Pointer definition
57 ;-----
58 .global  __STACK_END
59 .sect    .stack
60
61 ;-----
62 ; Interrupt Vectors
63 ;-----
64 .sect    ".reset"                ; MSP430 RESET Vector
65 .short   RESET
66 .end
67

```

**Figure 1. MSP430 Assembly Code for Count Character Program (Lab4\_D1.asm)**

## 2.2 Creating an Application Project


Please consult the first tutorial on how to create a new workspace and project.

Step 1. Choose Project>Create New CCS Project. This time you need to select Empty Assembly-only Project and save it as Lab4\_D1.

Step 2. Download and copy Lab4\_D1.asm into the project and delete main.asm by right clicking it and selecting delete or by clicking the delete key.


Step 3. Right-Click the project and select Properties then set all options as you did in Lab 1.

Step 4. Select the source file Lab4\_D1.asm and compile it. (Right-Click->Build). Click on the list file Lab4\_D1.lst to see report generated by the assembler.

Step 5. Link the program . The application is ready for debugging.

## 2.3 Program Simulation and Debugging

In this section we will discuss how to debug the program.

Step 1. Click the  icon.

Step 2. Step through the program step by step (use F6). Observe the Disassembly, Register View, and Memory View windows. Answer the following questions.

*What is the starting address of the program?*

*How many clock cycles does each instruction take to execute? How do the clock cycles taken by each instruction depend on addressing mode?*

**Hint:** To observe the number of clock cycles taken by your program, make sure to enable Clocks (checkboxes for Run>Clocks>Enable, Run>Clocks>Show should be set). In the bottom right corner of the CCStudio near the Stopwatch icon you will see the number that displays the current clock cycle.

Step 3. Observe the contents of memory location and registers as you step through the program. What is the content of the memory location at the address 0xFFFE?

What is the address of the special-purpose register P1OUT? Monitor the contents of relevant registers and address in the address space as you walk through your program. Set breakpoints to move easier through your program.






Figure 2. Code Composer Studio Debugging Window

### 3 References

It is important to be comfortable with the MSP430 instruction set and addressing modes. Your class readers and the MSP430 user's guide contain invaluable information that you will need to understand in order to effectively program the MSP430. Please read the following:

- [MSP430 ISA](#)
- [MSP430 Assembly Language Programming](#)
- Chapter 5, pages 119 - 146 in John H. Davies' MSP430 Microcontroller Basics
- Sections 6.4 and 6.5 in the MSP430F5529 User's Guide (<http://www.ti.com/lit/ug/slau208q/slau208q.pdf>) on Addressing modes and the instruction set (pages 196 - 341)